

Pratica con il colore

Una puntata pratica nella quale tratteremo alcune problematiche di uso del colore.

Tratteremo le problematiche base, proponendo alcune soluzioni in IBM Basic (versione 3.2 o successive per poter utilizzare, in alcuni esempi, una macchina classe IBM con scheda EGA e con Monitor avanzato).

Anticipiamo un discorsetto sul rapporto tra problematiche di Computer Grafica e soluzioni a tali problematiche in dipendenza dello strumento che si usa. In particolare se per fare Computer Grafica si usa un linguaggio occorrerà utilizzare istruzioni elementari ed assemblarle in programmi, se si usa un pacchetto grafico, occorrerà utilizzare le sue funzionalità.

In tal caso il risultato ottenuto dipende dalla «dotazione» di tale pacchetto.

Per esemplificare tale discorso rimaniamo a due illustrazioni (figg. 12,14) che rappresentano due modi di risolvere uno stesso problema di Business Graphic Tridimensionale.

La prima figura è ottenuta per mezzo di un programma Basic, non molto com-

plicato, ma con il quale ciascun elemento dell'Istogramma viene costruito, faccia per faccia e colorato faccia per faccia. La seconda è ottenuta per mezzo del prodotto Integrato Open Access 2 (prova su MC n. 60) che dispone di una specifica funzionalità grafica chiamata 3D.

È evidente che dato un risultato da ottenere utilizzando un linguaggio attrezzato per la grafica lo si può comunque ottenere. Se si dispone di un pacchetto che permette lo stesso risultato, lo si ottiene molto più semplicemente.

Trattiamo dunque alcuni temi base in Basic, come al solito con un po' di trattazione teorica, e un po' di pratica rappresentata da programmi esemplificativi e relativo output video.

Gerarchie di figure

Una immagine generica è composta di una serie di elementi. Ciascuno di questi elementi ha una sua forma o un suo colore o una sua posizione rispetto agli altri.

Nel caso della grafica tridimensionale entra in gioco anche la terza dimensione, e quindi entra in gioco anche una gerarchia di figure, rappresentata dalla posizione di queste ultime rispetto all'osservatore.

In questo caso alcune figure possono entrare in conflitto tra di loro nel senso che appaiono all'osservatore posti nella stessa posizione. Entra in gioco la gerarchia delle figure per cui si «vedono» le figure che stanno «davanti» mentre quelle che stanno indietro o si vedono solo parzialmente o non si vedono affatto.

Per esemplificare prendiamo un programmino, scritto in Basic IBM, che disegna tre rettangoli ciascuno dei quali occupa parzialmente la stessa posizione degli altri due. Il calcolo combinatorio indica in sei le possibili gerarchie e le vediamo nella figura 2.


Il listato è relativo non allo stesso output, ma lo semplifica permettendo un input di tre valori numerici (i numeri 1,2,3) che stabiliscano la gerarchia dei tre rettangoli.

Viene utilizzato lo SCREEN 1 che

```
100 REM SOVRAPPOSIZIONE BASE FIG01
110 SCREEN 1:CLS:WINDOW( 0,0)-(120,80)
120 FOR I=1 TO 3:READ X(I),Y(I),L(I),H(I),C(I):NEXT I
130 LOCATE 23,1 :INPUT A,B,C
140 LINE (X(A),Y(A))-(X(A)+L(A),Y(A)+H(A)),C(C(A)),BF
150 LINE (X(B),Y(B))-(X(B)+L(B),Y(B)+H(B)),C(C(B)),BF
160 LINE (X(C),Y(C))-(X(C)+L(C),Y(C)+H(C)),C(C(C)),BF
170 GOTO 130
180 DATA 10,10,60,40,1,60,20,60,40,2,40,40,40,40,3
```

Figura 1 - Rettangoli Sovrapposti - Listato. L'istruzione LINE (X1,Y1)-(X2,Y2), C,BF traccia tra le coordinate indicate un rettangolo pieno di colore C, indipendentemente da cosa esista sullo sfondo. In pratica il rettangolo si sovrappone ai precedenti.

Figura 2 - Rettangoli Sovrapposti - Output. Sono visualizzate le sei combinazioni possibili di graduatoria tra tre elementi. Il disegno di ogni rettangolo si sovrappone a quelli tracciati in precedenza.



permette di scegliere tre colori oltre a quello dello sfondo. I dati relativi ai tre rettangoli sono memorizzati nei vettori X,Y,L,H e il rispettivo colore nel vettore C (X,Y vertice superiore sinistro, L larghezza e H altezza).

Vengono chiesti i tre valori A,B,C, che debbono essere, dati in qualsiasi ordine, 1,2,3, e sulla base di questo input vengono visualizzati i tre rettangoli.

L'estrema semplicità di tale programma sta nel fatto che la funzione LINE..., C,BF traccia un rettangolo pieno, provvedendo a riempirlo del colore voluto, indipendentemente dal contenuto dello sfondo.

La stessa semplicità non è possibile quando si voglia colorare una figura di forma qualsiasi, su uno sfondo composto da figure e linee di forma e colore qualsiasi.

L'istruzione che si utilizza per riempire di colore una figura è, parliamo sempre di BASICA IBM o GWBASIC, PAINT (X,Y), C. Ma il suo funzionamento è «hardware» nel senso che è il PAINT che si va a cercare la figura da riempire identificandone i contorni identificando direttamente sulla Video Memory i Pixel di colore primo piano.

La sintassi completa della Funzione PAINT è:

PAINT (X,Y),C,L,S

X,Y coordinate del punto da cui far partire il riempimento.

C è il colore, che può essere indicato con un numero (compreso tra quelli possibili per la modalità che si sta usando). Può anche essere indicato come sequenza di CHR\$(N) per comporre dei colori costruiti come «miscela» di colori possibili (tecnica della retina).

L è il colore, presente sullo schermo, da considerare come bordo entro cui riempire.

S sfondo, serve per annullare lo sfondo in caso di riempimento di un'area con il metodo del retino, su cui già sia presente un altro retino.

Come si vede una funzione poliedrica tutta da verificare. Noi ci proponiamo di utilizzarla nella modalità più semplice cercando di raggiungere comunque i nostri obiettivi. La utilizzeremo solo nella forma PAINT (X,Y),C.

I limiti della funzione Paint

Sulla Video Memory non esiste il concetto di figura, e quindi il concetto di riempimento non può essere riferito ad una specifica figura. La conseguenza di ciò è che non si riesce in maniera semplice a tracciare figure e a colorarle.




Figure 3.4 - Limiti della Istruzione Paint. L'istruzione PAINT lavorando direttamente sulla Video Memory, non è in grado di riconoscere la figura sulla quale deve agire, che potrebbe essere sporca di linee e figure sottostanti. Il problema può essere risolto pulendo prima con un riempimento «sicuro» la stessa area.

Figura 5
Riempimento
Completo
Listato. L'uso del
trucco citato comporta
il fastidio di dover
riempire più volte la
stessa figura, le prime
per pulirla l'ultima
per colorarla.

```
100 REM SOVRASCITTURA seconda versione FIG05
110 SCREEN 7:COLOR 1,0:CLS:RANDOMIZE TIMER
120 I=1+1:IF I=16 THEN I=1
130 R=RND*70+10:C=I:K=K+1:IF K>4 THEN K=1
140 X1=RND*200+60:Y1=RND*120+40
150 LOCATE 22,1:INPUT A$
160 ON K GOSUB 170,200,230:GOTO 120
170 REM cerchio
180 FOR G=0 TO C:CIRCLE (X1,Y1),R,G
190 PAINT (X1,Y1),G:NEXT G:RETURN
200 REM quadrato
210 FOR G=0 TO C:LINE (X1,Y1)-(X1+R,Y1+R),G,BF
220 PAINT (X1,Y1),G:NEXT G:RETURN
230 REM rombo
240 FOR G=0 TO C:PSET (X1-R,Y1),G:LINE-(X1,Y1+R),G
250 LINE-(X1+R,Y1),G:LINE-(X1,Y1-R),G
260 LINE-(X1-R,Y1),G:PAINT (X1,Y1),G:NEXT G:RETURN
```




Figura 6 - Curva Tridimensionale - Metodo Linee - Output. Ogni elemento quadrangolare (nello spazio) è visualizzato per mezzo del suo perimetro. L'effetto è di un solido rappresentato tramite il suo scheletro. Un solido geometrico viene rappresentato per mezzo dei suoi spigoli.



Figura 8 - Curva Tridimensionale - Metodo Aree - Output. Se invece di visualizzare «spigoli» si visualizzano «aree» e magari di colori differenti. L'effetto, a parità di oggetto rappresentato, e in questo caso vengono usati colori Random, da cui il nome del programma Arlecchino, è ben differente.

Nella figura 3 mostriamo gli effetti di un programma che calcola e traccia Quadrati, Cerchi e Rombi di dimensioni, posizioni e colori casuali. Come si vede in certi casi l'istruzione Paint è disturbata dalla presenza di elementi

di sfondo di forma e colore varie.

Nella figura 4 vediamo invece l'effetto dello stesso programma modificato per lavorare correttamente. Proponiamo il listato in figura 5.

Ogni figura viene riempita più volte,

mediante un loop che fa variare il colore dal valore 0 a quello finale voluto, le prime a scopo di pulizia, realizzata con un riempimento di colore sicuro che liberi l'interno della figura di linee e aree di sfondo, che poi comunque verrebbero ricoperte, la seconda per la colorazione vera e propria.

L'aver risolto il problema del riempimento delle aree di forma qualsiasi ci permette di adottare la stessa soluzione per problemi più impegnativi di grafica tridimensionale.

Grafica Tridimensionale Linee e Aree

Le due figure 6 e 8 rappresentano una stessa curva dello spazio, realizzata in due modi differenti. Il primo è una rappresentazione per linee (difficilissima e trattata più volte in questa stessa rubrica) e la seconda è una rappresentazione per aree.

La rappresentazione per linee consiste nel calcolare, mediante dei loop sulle grandezze X,Y, il corrispondente valore Z, poi nel tradurre il punto (X,Y,Z) trovato in coordinate schermo e nella visualizzazione del segmento schermo che unisce il punto appena calcolato con quello calcolato prima.

Il vantaggio di tale metodo è la ripetitività delle operazioni che permette di «rigirare» i loop per visualizzare due famiglie di curve parallele, tra loro perpendicolari utilizzando le stesse subroutine di calcolo del valore Z e la stessa subroutine per la traduzione da valori X,Y,Z in valori schermo.

L'unica accortezza da avere è quella di identificare la condizione di partenza di ciascuna linea perché non comporta il disegno del segmento di congiunzione al punto precedente. Data la sua brevità riportiamo anche il listato del programma (fig. 7).

```

100 REM TRIDI FIG07
110 SCREEN 9:COLDR 9,1:CLS:NI=18:XC=320:YC=200:P=3.14159:S=P/B
120 Q1=4:Q2=2:CC=COS(Q1):CD=SIN(Q1):SC=COS(Q2):SD=SIN(Q2)
130 FOR A=-NI TO NI:FOR B=-NI TO NI:GOSUB 190
140 IF B=-NI THEN PSET (XS,YS) ELSE LINE-(XS,YS)
150 NEXT B:NEXT A
160 FOR B=-NI TO NI:FOR A=-NI TO NI:GOSUB 190
170 IF A=-NI THEN PSET (XS,YS) ELSE LINE-(XS,YS)
180 NEXT A:NEXT B:END
190 AO=A*S:BO=B*S:R=SDR(AO*AO+BO*BO)/2
200 IF R=0 THEN C=15 ELSE CO=-5*SIN(3*R)/(R+.00001)
210 XS=(AO*CC+BO*CD)*30+XC:YS=(AO*SC+BO*SD+CO)*10+YC:RETURN

```

Figura 7 - Curva Tridimensionale - Metodo Linee - Listato. Ci permettiamo di presentare un listato analogo ad altri già pubblicati, ma oltre ad essere particolarmente corto rappresenta un programma «classico» della Computer Grafica.

```

100 REM ARLECCHINO FIG09
110 SCREEN 9:COLDR 1,9:CLS:NI=18:NJ=18:XC=320:YC=200:P=3.14159:SP=P/B
120 DIM XX(2*NI,2*NJ),YY(2*NI,2*NJ)
130 Q1=1:Q2=3:CC=COS(Q1):CD=SIN(Q1):SC=COS(Q2):SD=SIN(Q2)
140 FOR A=-NI TO NI:FOR B=-NJ TO NJ:AO=A*SP:BO=B*SP:A1=A+NI:B1=B+NJ
150 R=SDR(AO*AO+BO*BO)/2:CO=-5*SIN(3*R)/(R+.0001):GOSUB 270
160 XX(A1,B1)=XS:YY(A1,B1)=YS:NEXT B:NEXT A
170 CLS:FOR A=2*NI TO 1 STEP -1:FOR B=1 TO 2*NJ
180 X1=XX(A,B):Y1=YY(A,B):X2=XX(A-1,B):Y2=YY(A-1,B)
190 X3=XX(A-1,B-1):Y3=YY(A-1,B-1):X4=XX(A,B-1):Y4=YY(A,B-1)
200 K=K+1:IF K>15 THEN K=1
210 PSET (X1,Y1),K:LINE-(X2,Y2),K:LINE-(X3,Y3),K:LINE-(X1,Y1),K
220 X5=(X1+X2+X3)/3:Y5=(Y1+Y2+Y3)/3
230 LINE-(X1,Y1),K:LINE-(X4,Y4),K:LINE-(X3,Y3),K:LINE-(X1,Y1),K
240 X6=(X1+X3+X4)/3:Y6=(Y1+Y3+Y4)/3
250 PAINT (X5,Y5),K:PAINT (X6,Y6),K
260 NEXT B:NEXT A:END
270 XS=(AO*CC+BO*CD)*30+XC:YS=(AO*SC+BO*SD+CO)*10+YC:RETURN

```

Figura 9 - Curva Tridimensionale - Metodo Aree - Listato. Le difficoltà in termini di programmazione aumentano un bel po' rispetto al metodo di visualizzazione per linee. La difficoltà principale dipende dalla limitatezza della funzionalità di riempimento PAINT, in caso di aree irregolari.





Figura 10 - Hidden Line - Output. Il metodo più semplice di eliminazione delle linee nascoste, in caso di utilizzo del metodo «linee» nella rappresentazione di una superficie spaziale, è quello dei Massimi e dei Minimi, per mezzo del quale si sottopongono a test i valori delle coordinate schermo prodotte dai calcoli.

Figura 12 - Istogramma Tridimensionale in Basic - Output. Esempio di applicazione del metodo della sovrapposizione per eliminare non tanto le linee nascoste, quanto «il problema» delle linee nascoste.

Ricordiamo che in tale programma è riportata la routine fondamentale della grafica tridimensionale, opportunamente semplificata ed abbreviata, per mezzo della quale dati tre valori X,Y,Z si ottiene la coppia dei valori video. La routine è in riga 180-190.

Passiamo a ragionare in termini di aree.

Una superficie spaziale può essere rappresentata sia attraverso delle linee che la compongono che attraverso porzioni di area piane, che la approssimino. Il grado di approssimazione è tanto maggiore quanto maggiore è il numero delle aree e, inversamente, minore la loro dimensione.

Applichiamo immediatamente tale metodo alla figura precedente. Nel caso delle linee queste si possono visualizzare durante il calcolo. Nel caso delle aree invece ogni areola, approssimata ad un quadrangolo di forma irregolare, è individuata da quattro punti, e quindi conviene dapprima calcolare tutti i punti e poi visualizzare quattro a quattro quelli contigui.

Però se nella curva spaziale quattro punti contigui possono essere approssimati ad un quadrilatero, non è detto che questo quadrilatero sia complanare (nello spazio tre punti identificano un piano, il quarto punto non è detto che sia nello stesso piano).

Inoltre gli stessi quattro punti, portati sul piano video non è detto che formino un quadrilatero convesso, potrebbero infatti «incrociarsi».

Per risolvere questo problema riduciamo il quadrilatero in due triangoli (con un angolo in comune) e non avremo più problemi spaziali in quanto come detto per tre punti passa un solo piano. In tal modo potremo calcolare i punti, tracciare i due triangoli che compongono una singola areola e colorarli, riempendoli con il sistema il-

lustrato in precedenza.

Il risultato del lavoro è in figura 8. A commento del listato di figura 9 alcune precisazioni.

Nel caso di disegno di superfici nello spazio calcolate e visualizzate per

mezzo di loop viene molto semplificato il problema delle gerarchie, infatti basta cominciare il disegno «da dietro» e automaticamente le varie areole si sovrappongono alle precedenti.

Noti i tre punti che individuano il

```

100 REM HIDDEN FIG11
110 SCREEN 9:CLS:NI=28:NJ=28:XC=320:YC=200:P=3.14159:S=P/12
120 O1=4:O2=2:CC=COS(O1):CD=SIN(O1):SC=COS(O2):SD=SIN(O2)
130 DIM Y(639):FOR I=0 TO 639:Y(I)=349:NEXT I
140 FOR A=NI TO NJ:K=K+1:IF K>15 THEN K=1
150 FOR B=NJ TO -NJ STEP -1:GOSUB 180
160 IF B=NJ THEN XP=XB:YP=YB ELSE GOSUB 210:XP=XB:YP=YB
170 NEXT B:NEXT A:END
180 AO=A*S:BO=B*S:R=SDR(AO*AO+BO*BO)/2
190 IF R=0 THEN C=15 ELSE C=0=-S*SIN(3*R)/(R+.0001)
200 XS=(AO*CC+BO*CD)*30+XC:YS=(AO*SC+BO*SD+C)*10+YC:RETURN
210 DX=XB-XP:DY=YB-YP:IF ABS(DY)>ABS(DX) THEN 240
220 DD=DY/DX:T=1:IF DX<0 THEN T=-1
230 FOR X=XP TO XS STEP T:Y=YP+(X-XP)*DD:GOSUB 260:NEXT X:RETURN
240 DD=DX/DY:T=1:IF DY<0 THEN T=-1
250 FOR Y=YP TO YS STEP T:X=XP+(Y-YP)*DD:GOSUB 260:NEXT Y:RETURN
260 IF Y<Y(X) THEN PSET(X,Y),K:Y(X)=Y:RETURN ELSE RETURN

```

Figura 11 - Hidden Line - Listato. L'argomento Linee Nascoste è stato trattato nei numeri 11 e 12 di MC microcomputer. Notare nel listato l'uso del vettore V(X) che memorizza il più alto valore Y visualizzato per ciascun valore X permesso dallo schermo di uscita.

```

100 REM ISTOGRAMMA BASIC FIG13
110 SCREEN 8:COLOR 1,3:CLS:LINE (0,0)-(639,199),,B
120 R=7:C=6:XC=240:YC=180:H=100:LD=280:LS=192:SD=LD/R:SS=LS/C:SH=B:P=.5
130 TD=P*SD:TS=P*SS:TH=P*SH:XA=XC-SS*C:YA=YC-SH*C:XB=XC+SD*R:YB=YC-SH*R
140 PSET (XA,YA-H):LINE-(XA,YA):LINE-(XC,YC):LINE-(XB,YB):LINE-(XB,YB-H)
150 FOR I=0 TO R-1:READ A#:LOCATE 24-I,35+I*5:PRINT A#:NEXT I
160 FOR I=0 TO C-1:READ A#:LOCATE 24-I,22+I*4:PRINT A#:NEXT I
170 FOR I=6 TO 0 STEP -1:FOR J=5 TO 0 STEP -1
180 A=(I+3)*(J+2)+4:GOSUB 190:NEXT J:NEXT I:END
190 REM Box sinistra/destra/superiore
200 X1=XC+I*SD-J*SS:Y1=YC-I*SH-J*SH-A:X2=X1:Y2=Y1+A
210 X3=X2-TS:Y3=Y2-TH:X4=X1-TS:Y4=Y1-TH:X5=X4:Y5=Y4:Q=4:GOSUB 240
220 X3=X2+TD:Y3=Y2+TH:X4=X1+TD:Y4=Y1+TH:Q=6:GOSUB 240
230 X2=X5:Y2=Y5:X3=X2+TD:Y3=Y2-TH:X4=X4:Y4=Y4:Q=1:GOSUB 240:RETURN
240 FOR C=1 TO Q : REM Rombo
250 PSET(X1,Y1),C:LINE-(X2,Y2),C:LINE-(X3,Y3),C:LINE-(X4,Y4),C:LINE-(X1,Y1),C
260 X9=(X1+X3)/2:Y9=(Y1+Y3)/2:PAINT (X9,Y9),C:NEXT C:RETURN
270 DATA Francia,Belgio,Olanda,Germania,Italia,Gran Bretagna,Spagna
280 DATA "Pere","Mele","Noci","Pesche","Arance","Banane"

```

Figura 13 - Istogramma Tridimensionale in Basic - Listato. L'esiguità del listato dà la misura di come un problema apparentemente complesso possa essere ridotto all'osso. È evidente che sono state evitate totalmente sia le scritte che qualsiasi fronzolo estetico.





Figure 14, 15 - Business Graphic Tridimensionale con Open Access. L'unico prodotto di tipo integrato che disponga di grafica tridimensionale è Open Access che permette, come uscita dei dati numerici contenuti nel tabellone, rappresentazioni pseudotridimensionali, sia di tipo Barre, che di tipo Superficie.

vertice dell'areola il metodo più semplice per cercarsi un punto interno dal quale far partire il riempimento consiste nel calcolare il baricentro dell'areola mediando le due terne di coordinate.

Tale metodo va sempre bene, meno quando si tratta di un triangolo talmente schiacciato che il baricentro così calcolato giace su uno dei lati del triangolo. Tale caso andrebbe testato ed escluso dalla routine di calcolo.

Linee nascoste

Anche i lettori più distratti si saranno accorti che in pratica il metodo suggerito è un metodo di eliminazione delle linee nascoste, uno di quei metodi semplificati che agiscono sul risultato della visualizzazione e non sui dati geometrici del disegno e del punto di osservazione.

In Computer Grafica tradizionale, in particolare nella progettazione industriale tramite computer tale metodo si chiama Solid Modelling. Rispetto alla classica rappresentazione per linee dell'oggetto (ad esempio una carrozzeria) si ha una rappresentazione solida ottenuta per mezzo di aree omogeneamente colorate.

Il vantaggio, oltre a quello di eliminare l'ambiguità di interpretazione data dalle linee nascoste sta nel fatto di poter utilizzare nella fase progettuale anche i colori pieni sia come elemento che dà corpo all'immagine, mediante l'effetto chiaroscuro, sia come elemento di progettazione, per scegliere accostamenti di colori.

Visto che trattiamo le linee nascoste riproponiamo rapidamente il classico metodo dei Massimi per mezzo del quale ciascun punto da visualizzare viene sottoposto a test. Il test consiste nel verificare se per quel valore X del punto è già stato visualizzato un punto con valore Y superiore.

Se lo è stato il punto è nascosto, se non lo è stato il punto non è nascosto e va visualizzato. In tal caso il vettore $Y(X)$ per quel particolare X assume il nuovo valore Y. Listato in figura 11 e output in figura 10. In tale metodo va innanzitutto inizializzato il vettore dei massimi con un apposito loop, poi ogni segmento va scomposto nei suoi punti perché va designato punto per punto. Inoltre si deve lavorare direttamente in coordinate video (nel nostro caso 640 per 350 pixel dello SCREEN 9 EGA del PC IBM) in quanto il test di visibilità può essere eseguito esclusivamente sulle coordinate video.

Altra difficoltà insita nel sistema di visualizzazione per punti sta nello scegliere in quale senso eseguire il loop, se nel senso della X o in quello della Y. In altre parole il segmento tra il punto $P1(X1,Y1)$ e $P2(X2,Y2)$ si può eseguire calcolando le Y facendo variare la X tra $X1$ e $X2$ o viceversa.

Se il segmento è molto inclinato un incremento di un pixel nel senso X, può far variare la Y di molte unità rendendo la linea eccessivamente puntinata. Il problema si risolve facilmente testando l'inclinazione e scegliendo di conseguenza il senso del loop.

Business Graphic

Tornando al metodo delle aree esportiamolo nel campo della Business Graphic, per realizzare un Istogramma Tridimensionale, semplificando al solito il problema, limitandolo cioè alla visualizzazione delle sole Barre ignorando Titoli, Legende e scritte varie.

Anche in questo esempio occorre partire dal fondo, in modo che le barre via via si sovrappongano a quelle di livello inferiore e che i colori ciclino da 1 a quello scelto per poter effettivamente riempire i vari quadrangoli.

Per rendere ancora più corto il lista-

to (fig. 13 e output in fig. 12) abbiamo calcolato mediante formula di calcolo il valore della coordinata verticale delle singole barre.

Quindi in caso di implementazione del programma per un utilizzo su dati qualsiasi questi vanno letti con apposite istruzioni READ e DATA, oppure con funzioni di INPUT., oppure da file dati.

Sono invece parametrizzati i numeri delle righe e delle colonne (R,C di riga 120) per cui entro un range accettabile è possibile variare il numero delle righe e delle colonne per adattarlo alle varie esigenze.

Di tale ultima problematica esaminiamo anche la versione Open Access 2. Questo particolare pacchetto integrato dispone infatti di una opzione di Business Graphic di tipo tridimensionale.

Interessante di questa funzionalità, come risulta anche dalla prova su MC numero 60, è la sua semiautomaticità per mezzo della quale è possibile in automatico disegnare le BARRE, mentre scritte, orientamenti, colori del disegno vanno scelti con apposite funzioni che agiscono su barre di opzioni.

Su tale barra si può agire anche con il mouse alla solita maniera, ad esempio andandosi a prelevare il colore voluto tra quelli disponibili.

Altra funzionalità permessa è la surface, ovvero il disegno di superfici spaziali. Anche OPEN ACCESS come i vari LOTUS 123, FRAMEWORK, ENABLE, ecc. permettono di graficare dati presenti nel tabellone elettronico (per OA che lavora per moduli (vedi articolo sugli Spreadsheet) Tabellone e Business Graphic sono un unico modulo).

Vediamo nelle due figure finali due disegni realizzati con Open Access tipologia 3D e Surface.


ANCHE L'OCCHIO VUOLE LA SUA PARTE DI TECNOLOGIA



Più alte sono le prestazioni, migliore e più avanzato deve essere lo schermo. Ecco perchè l'ASEM ha recentemente adottato i nuovissimi monitor e terminali a schermo piatto della Wyse Technology.

Tutta la vasta gamma dei monitor e terminali Wyse (14 pollici con fosforo verde, ambra o bianco) è dotata di uno schermo nero con sottili incisioni che evitano fastidiosi riflessi e di una matrice molto definita che consente elevate risoluzioni.

Accendete i vostri schermi. Tutte le emulazioni (UNIX, DEC, DATA GENERAL, IBM RT PC, ecc.) sono possibili.



Comunicare

WYSE

DISTRIBUZIONE



Asem spa Buia (UD) tel. 0432/962118 r.a.
Telex 450608 Fax 0432/960282

Scrivi 400 Mb, si rimuove come un floppy si usa come un Winchester è un disco ottico **Optotech**

I drive Ottici Optotech scrivono i dati su di una cartuccia removibile da 5,25".

Pratica e facile da usare quanto un Floppy, ogni cartuccia ha una capacità di più di 400 Megabytes (200 per faccia).


I drive Optotech si usano come un qualsiasi

Winchester e permettono di archiviare economicamente una massa di informazioni illimitata.

I dati immessi sono leggibili in qualsiasi momento ma non più cancellabili, garantendo una assoluta sicurezza di archiviazione.

Optical Disk drive

— Capacità formattata 202,4 Mb per faccia



— Tecnologia di lettura durante la scrittura, con controllo in tempo reale della corretta registrazione.

SCSI Controller

Previsto per collegare da uno a quattro Drive alla porta SCSI.

PC Controller

Previsto per collegare da uno a quattro Drive a IBM PC, XT, AT e compatibili.

**Soluzioni pronte
all'uso per:
IBM PC, XT, AT,
Olivetti;
Compatibili;**

Apple Macintosh

Per maggiori informazioni sui prodotti distribuiti dalla Contradata, telefonate allo 039/737015 o scrivete a Contradata s.r.l., via Monte Bianco 4, 20052 Monza (MI), telex 314112 MANCEN - fax 039-735276 G3.



contradata

PER COMPUTER CHE NON HANNO TEMPO DA PERDERE